
      SUBROUTINE QUEST(NOBS,VREF,WOBS,DEV,QUIBBL,FIBBL,QUACC,NEWT,

     1                 IMETH,QOPT,COVAR,SIGTOT,RESIDU,TASTE,ICON,IOPT)

C

C-----VERSION DATE- JANUARY 19, 1987 ----------------------------------C

C                                                                      C

C-----FUNCTION-- QUEST COMPUTES THE QUATERNION OF THE OPTIMAL          C

C                ROTATION CARRYING A SET OF REFERENCE UNIT             C

C                VECTORS INTO A SET OF CORRESPONDING OBSERVATION       C

C                UNIT VECTORS AND THE COVARIANCE MATRIX OF THE         C

C                QUATERNION. THE ROTATION MINIMIZES A QUADRATIC        C

C                LOSS FUNCTION                                         C

C                                                                      C

C                ANALYSIS FOR THIS ALGORITHM IS GIVEN IN               C

C                   M. D. SHUSTER, APPROXIMATE ALGORITHMS FOR          C

C                   FAST OPTIMAL ATTITUDE COMPUTATION, PAPER NO.       C

C                   78-1249, AIAA GUIDANCE AND CONTROL CONFER-         C

C                   ENCE, PALO ALTO, CALIF., AUGUST 1978               C

C                                                                      C

C-----ARGUMENT LIST--                                                  C

C                                                                      C

C     NAME          TYPE I/O     DESCRIPTION                           C

C     ----          ---- ---     -----------                           C

C                                                                      C

C     NOBS          I*4   I      NUMBER OF OBSERVATIONS                C

C                                                                      C

C     VREF(3,NOBS)  R*4   I      REFERENCE UNIT VECTORS                C

C                                                                      C

C     WOBS(3,NOBS)  R*4   I      OBSERVATION UNIT VECTORS              C

C                                                                      C

C     DEV(NOBS)     R*4   I      STANDARD DEVIATIONS PER AXIS OF       C

C                                UNIT VECTOR MEASUREMENTS (IN RAD)     C

C                                QUEST WILL SET AN ERROR CODE (ICON=3) C

C                                IF ANY OF THE STANDARD DEVIATIONS IN  C

C                                THE COMPUTATION ARE SET EQUAL TO ZERO.C

C                                                                      C

C     QUIBBL        R*4   I      CRITICAL VALUE FOR TESTING QUALITY    C

C                                OF QUATERNION COMPUTATION             C

C                                                                      C

C     FIBBL         R*4   I      CRITICAL VALUE FOR TESTING CONDITION  C

C                                OF OBSERVABILITY MATRIX               C

C                                                                      C

C     QUACC         R*4   I      DESIRED ACCURACY IN COMPUTING         C

C                                OVERLAP EIGENVALUE (ALAM)             C

C                                                                      C

C     NEWT          I*4   I      NUMBER OF NEWTON-RAPHSON ITERATIONS   C

C                                IN QUATERNION COMPUTATION             C

C                                                                      C

C     IMETH         I*4   I      ALGORITHM SELECTION FLAG              C

C                                (IF IMETH = 0, THE METHOD             C

C                                OF SEQUENTIAL ROTATIONS               C

C                                IS NOT IMPLEMENTED                    C

C                                                                      C

C     QOPT(4)       R*4   O      OPTIMAL QUATERNION (DEFAULT VALUE     C

C                                = (999.,999.,999.,999.))              C

C                                                                      C

C     COVAR(3,3)    R*4   O      COVARIANCE MATRIX (UNITLESS)          C

C                                (DEFAULT VALUE = 999. IN EVERY        C

C                                ELEMENT)                              C

C                                                                      C

C     SIGTOT        R*4   O      TOTAL STANDARD DEVIATION OF           C

C                                OBSERVATION MEASUREMENTS              C

C                                                                      C



C     RESIDU        R*4   O      OVERLAP DEFECT (=1.0 - OVERLAP)       C

C                                                                      C

C     TASTE         R*4   O      STATISTICAL PARAMETER (IF VALUES OF   C

C                                DEV(NVEC) ARE CORRECT, TASTE WILL     C

C                                POSSESS A CHI-SQUARE DISTRIBUTION     C

C                                WITH 2*NVEC DEGREES OF FREEDOM)       C

C                                                                      C

C     ICON          I*4   O      CONDITION CODE                        C

C                                 ICON=0 -- NO ERRORS OR POOR CONDI-   C

C                                           TIONS  DETECTED            C

C                                 ICON=1 -- OBSERVATION COUNT CHECK    C

C                                           (FEWER THAN TWO            C

C                                           OBSERVATIONS)              C

C                                 ICON=2 -- NEGATIVE WEIGHT CHECK      C

C                                           (NEGATIVE WEIGHT DETECTED) C

C                                 ICON=3 -- INFINITE WEIGHT CHECK      C

C                                           (AT LEAST ONE STANDARD     C

C                                           DEVIATION IS ZERO)         C

C                                 ICON=4 -- NORM CHECK                 C

C                                           (AT LEAST ONE REFERENCE    C

C                                           OR OBSERVATION VECTOR      C

C                                           HAS ZERO NORM)             C

C                                 ICON=5 -- OBSERVABILITY CHECK        C

C                                           (OBSERVABILITY MATRIX ILL- C

C                                           CONDITIONED)               C

C                                 ICON=6 -- FIRST QUATERNION FAILURE   C

C                                           CHECK (QUATERNION INACCU-  C

C                                           RATE AND IMETH=0)          C

C                                 ICON=7 -- SECOND QUATERNION FAILURE  C

C                                           CHECK (ALL FOUR QUATERNION C

C                                           COMPUTATIONS FAILED)       C

C                                                                      C

C     IOPT          I*4   O      OPTIMIZATION FLAG                     C

C                                 IOPT=0 -- QUATERNION COMPUTATION     C

C                                           SUCCEEDED ON FIRST COM-    C

C                                           PUTATION                   C

C                                                                      C

C                                 IOPT=1 -- A SUCCESSFUL QUATERNION    C

C                                           COMPUTATION WAS OBTAINED   C

C                                           BY METHOD OF SEQUENTIAL    C

C                                           ROTATIONS                  C

C                                 IOPT=2 -- ALL QUATERNION COMPUTATIONSC

C                                           FAILED. BEST QUATERNION IS C

C                                           OUTPUT.                    C

C                                                                      C

C-----EXTERNAL REFERENCES-- QUACOV - COMPUTES QUATERNION               C

C                                    COVARIANCE MATRIX                 C

C                                                                      C

C                           QUOMP  - COMPUTES UNNORMALIZED QUATERNION, C

C                                    OVERLAP, COMPUTATION QUALITY      C

C                                    PARAMETER                         C

C                                                                      C

C                           VEX    - COMPUTES A NORMALIZED VECTOR      C

C                                                                      C

C-----DESIGNER/PROGRAMMER--M. D. SHUSTER  C.S.C. 1/17/1979             C

C                                                                      C

C-----MODIFICATIONS-- F. L. MARKLEY GSFC CODE 552.2 1/19/87            C

C                     REMOVE LIMITATION ON NOBS.LT.10.  ALSO TEST FOR  C

C                     DEV**2=0 INSTEAD OF DEV=0 TO AVOID POSSIBLE      C

C                     REAL UNDERFLOW FOLLOWED BY DIVISION BY ZERO.     C

C                                                                      C

C-----Comment-- M. D. SHUSTER  4/03/2005                               C

C               Since 1981 IT HAS BEEN KNOW THAT, PROVIDED THE VALUES  C

C               OF DEV(NVEC) ARE CORRECT, TASTE WILL POSSESS A         C

C               CHI-SQUARE DISTRIBUTION WITH (2*NVEC-3) DEGREES OF     C

C               FREEDOM.                                               C

C                                                                      C

C               FURTHER INFORMATION ON THE QUEST ALGORITHM IS          C



C               AVAILABLE IN M. D. SHUSTER AND S. D. OH, THREE-AXIS    C

C               ATTITUDE DETERMINATION FROM VECTOR OBSERVATIONS,       C

C               JOURNAL OF GUIDANCE AND CONTROL, VOL. 4, NO. 1,        C

C               JAN-FEB 1981, PP. 70-77.                               C

C                                                                      C

C----------------------------------------------------------------------C

C                                                                      C

C                                                                      C

      REAL*8 V(3),W(3),SIG,SIG2,WT,QUALM,QNORM,DLAM,VNORM,WNORM,

     1       SIGTM2,SIGT2,SIGT,FIDDL,QUETCH,F(3,3),B(3,3),P(3,3),QUIP,

     2       Q(4),ALAM,QUAL,QQ(4,4),SALAM(4),SQUAL(4)

C

      DIMENSION VREF(3,*),WOBS(3,*),DEV(*),QOPT(4),COVAR(3,3)

C

C     PERFORM OBSERVATION COUNT CHECK

C

      IF(NOBS.LT.2) THEN

         ICON = 1

         GO TO 999

      ENDIF

C

C     SET DEFAULT VALUES FOR QUATERNION, COVARIANCE, CONDITION CODE,

C     AND OPTIMIZATION FLAG

C

      DO 20 I=1,3

      DO 10 J=1,3

      COVAR(I,J) = 999.

   10 CONTINUE

      QOPT(I) = 999.

   20 CONTINUE

      QOPT(4) = 999.

C

      ICON = 0

C

      IOPT = 0

C

C     CONVERT STANDARD DEVIATIONS TO REAL*8 AND PERFORM WEIGHT CHECKS. 

C     IF OK, COMPUTE TOTAL STANDARD DEVIATION.  OTHERWISE,

C     SET CONDITION CODE AND RETURN.

C

      SIGTM2 = 0.D0

C

      DO 80 I=1,NOBS

      IF(DEV(I).LT.0.) THEN

         ICON = 2

         GO TO 999

      ENDIF

      SIG = DEV(I)

      SIG2 = SIG**2

      IF(SIG2.EQ.0.D0) THEN

         ICON = 3

         GO TO 999

      ENDIF

      SIGTM2 = SIGTM2 + 1.D0/SIG2

   80 CONTINUE

C

      SIGT2 = 1.D0/SIGTM2

      SIGT  = DSQRT(SIGT2)

      SIGTOT = SIGT

      FIDDL = FIBBL

      QUETCH = QUACC

      QUIP = QUIBBL

C

C     CONVERT VECTORS TO REAL*8 AND PERFORM NORM CHECKS.  

C     IF OK COMPUTE WEIGHTS, OBSERVABILITY MATRIX, AND ATTITUDE 

C     PROFILE MATRIX.  OTHERWISE, SET CONDITION CODE AND RETURN.

C

      DO 100 I=1,3



      DO  90 J=1,3

      F(I,J) = 0.D0

      B(I,J) = 0.D0

   90 CONTINUE

      F(I,I) = 1.D0

  100 CONTINUE

C 

      DO 130 I=1,NOBS

      CALL VEX(VREF(1,I),VREF(2,I),VREF(3,I),V(1),V(2),V(3),VNORM)

      IF(VNORM.LE.0.D0) THEN

         ICON = 4

         GO TO 999

      ENDIF

      CALL VEX(WOBS(1,I),WOBS(2,I),WOBS(3,I),W(1),W(2),W(3),WNORM)

      IF(WNORM.LE.0.D0) THEN

         ICON = 4

         GO TO 999

      ENDIF

      SIG=DEV(I)

      WT = SIGT2/(SIG**2)

C

      DO 120 J=1,3

      DO 110 K=1,3

      F(J,K) = F(J,K) - WT*W(J)*W(K)

      B(J,K) = B(J,K) + WT*W(J)*V(K)

  110 CONTINUE

  120 CONTINUE

  130 CONTINUE

C

C

C     COMPUTE COVARIANCE MATRIX AND TEST PARAMETER

C

      CALL QUACOV(F,FIDDL,SIGT2,IDET,P)

C

C

C     TEST CONDITION OF OBSERVABILITY MATRIX. IF OK, CONTINUE,

C     OTHERWISE SET CONDITION CODE AND RETURN.

C

      IF(IDET.EQ.0)GO TO 140

      ICON = 5

      GO TO 999

C

C

C     COMPUTE UNNORMALIZED OPTIMAL QUATERNION AND TEST PARAMETER

C

  140 CALL QUOMP(B,NEWT,QUETCH,Q,ALAM,QUAL)

C

C

C     TEST QUATERNION COMPUTATION ACCURACY. IF OK, USE THIS FOR

C     COMPUTING THE NORMALIZED QUATERNION. OTHERWISE, KEEP TRYING

C

      IF(QUAL.GT.QUIP)GO TO 900

C

C

C     TEST WHETHER PROGRAM IS SPECIFIED TO USE METHOD OF SEQUENTIAL

C     ROTATIONS TO TRY TO FIND A BETTER RESULT. IF SO, IMPLEMENT

C     THIS METHOD. OTHERWISE, USE THE LAST COMPUTED QUATERNION.

C

      IF(IMETH.NE.0)GO TO 150

      ICON = 6

      GO TO 900

C

C

C     IMPLEMENT METHOD OF SEQUENTIAL ROTATIONS.  BY ROTATING REFERENCE

C     VECTORS AND PERFORMING THE CORRESPONDING INVERSE TRANSFORMATION

C     ON THE RESULTING QUATERNION, WE ATTEMPT TO OBTAIN A MORE ACCURATE

C     RESULT

C



C     LOAD FIRST QUATERNION, QUALITY PARAMETER, AND OVERLAP EIGENVALUE

C     FOR LATER REFERENCE.

C

  150 QQ(1,1) = Q(1)

      QQ(2,1) = Q(2)

      QQ(3,1) = Q(3)

      QQ(4,1) = Q(4)

C

      SALAM(1) = ALAM

      SQUAL(1) = QUAL

C

C

C     SET OPTIMIZATION FLAG TO SHOW THAT METHOD OF SEQUENTIAL ROTATIONS

C     IS BEING IMPLEMENTED.

C

      IOPT = 1

C

C     FIRST ROTATION.  UNNORMALIZED QUATERNION IS COMPUTED FOR A ROTA-

C     TION OF THE REFERENCE VECTORS THROUGH 180. DEGREES ABOUT THE

C     X-AXIS.  THE FOLLOWING STEPS ARE PERFORMED:

C

C     A NEW MEASUREMENT PROFILE MATRIX IS COMPUTED

C

      DO 160 I=1,3

      B(I,2) = -B(I,2)

      B(I,3) = -B(I,3)

  160 CONTINUE

C

C     THE UNNORMALIZED QUATERNION IS RECOMPUTED FOR THIS NEW VECTOR SET

C

      CALL QUOMP(B,NEWT,QUETCH,Q,ALAM,QUAL)

C

C     THE INVERSE TRANSFORMATION IS APPLIED TO THE QUATERNION, WHICH IS

C     LOADED IN THE SECOND COLUMN OF THE ARRAY QQ. LIKEWISE FOR THE

C     OTHER PARAMETERS.

C

      QQ(1,2) =  Q(4)

      QQ(2,2) = -Q(3)

      QQ(3,2) =  Q(2)

      QQ(4,2) = -Q(1)

C

      SALAM(2) = ALAM

      SQUAL(2) = QUAL

C

C     NOW WE TEST THIS QUATERNION FOR ACCURACY. IF IT IS OK, WE WILL

C     SET A FLAG TO DENOTE THAT THE SECOND COLUMN OF QQ WILL BE USED

C     TO COMPUTE THE NORMALIZED OPTIMAL QUATERNION AND PROCEED TO

C     THAT PART OF THE COMPUTATION. OTHERWISE, WE TRY THE NEXT

C     ROTATION.

C

      IF(QUAL.LT.QUIP)GO TO 170

C

      IQ = 2

      GO TO 800

C

C

C     WE NOW TRY A ROTATION OF THE REFERENCE VECTORS THROUGH 180. DEG

C     ABOUT THE Y-AXIS AND REPEAT THE SAME STEPS. NOTE THAT IN COMPUT-

C     ING THE NEW MEASUREMENT PROFILE MATRIX WE MUST TAKE ACCOUNT OF

C     PREVIOUS ROTATIONS.

C

  170 DO 180 I=1,3

      B(I,1) = -B(I,1)

      B(I,2) = -B(I,2)

  180 CONTINUE

C

      CALL QUOMP(B,NEWT,QUETCH,Q,ALAM,QUAL)

C



      QQ(1,3) =  Q(3)

      QQ(2,3) =  Q(4)

      QQ(3,3) = -Q(1)

      QQ(4,3) = -Q(2)

C

      SALAM(3) = ALAM

      SQUAL(3) = QUAL

C

      IF(QUAL.LT.QUIP)GO TO 190

C

      IQ = 3

      GO TO 800

C

C

C     NOW WE TRY A ROTATION OF THE REFERENCE VECTORS ABOUT THE Z-AXIS

C

  190 DO 200 I=1,3

      B(I,2) = -B(I,2)

      B(I,3) = -B(I,3)

  200 CONTINUE

C

      CALL QUOMP(B,NEWT,QUETCH,Q,ALAM,QUAL)

C

      QQ(1,4) = -Q(2)

      QQ(2,4) =  Q(1)

      QQ(3,4) =  Q(4)

      QQ(4,4) = -Q(3)

C

      SALAM(4) = ALAM

      SQUAL(4) = QUAL

C

      IF(QUAL.LT.QUIP)GO TO 210

C

      IQ = 4

      GO TO 800

C

C

C     STATEMENT 210 IS REACHED IF THE METHOD OF SEQUENTIAL ROTATIONS

C     FAILED TO FIND AN ACCEPTABLE QUATERNION. IF THIS IS A FREQUENT

C     OCCURENCE, IT PROBABLY INDICATES THAT QUACC, NEWT, AND QUIBBL

C     HAVE NOT BEEN CHOSEN PROPERLY. QUEST NOW SEARCHES FOR THE BEST

C     QUATERNION AND SETS THE OPIMIZATION FLAG TO 2.

C

  210 IOPT = 2

      ICON = 7

      IQ = 1

      QUALM = 0.D0

C

      DO 220 I=1,4

      IF(SQUAL(I).GT.QUALM) IQ = I

      IF(IQ.EQ.I) QUALM = SQUAL(I)

  220 CONTINUE

C

C

C     CHOOSE BEST QUATERNION AND OVERLAP EIGENVALUE

C

  800 DO 230 I=1,4

      Q(I) = QQ(I,IQ)

  230 CONTINUE

C

      ALAM = SALAM(IQ)

C

  900 DLAM = 1.D0 - ALAM

C

      QNORM = DSQRT(Q(1)**2 + Q(2)**2 + Q(3)**2 + Q(4)**2)

C

C     RENORMALIZE QUATERNION. CONVERT OUTPUT ARRAYS TO REAL*4

C



      DO 920 I=1,3

      DO 910 J=1,3

      COVAR(I,J) = P(I,J)

  910 CONTINUE

      QOPT(I) = Q(I)/QNORM

  920 CONTINUE

      QOPT(4) = Q(4)/QNORM

C

C     COMPUTE QUALITY PARAMETERS

C

      RESIDU = DLAM

C

      TASTE =  2.D0*DLAM/SIGT2

C

C

  999 RETURN

      END

C = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = =

      SUBROUTINE QUOMP(B,SIGTM2,NEWT,QUETCH,Q,ALAM,QUAL)

C

C-----VERSION DATE- JUNE 26, 1990 -------------------------------------C

C                                                                      C

C-----FUNCTION-- QUOMP COMPUTES AN UNNORMALIZED QUATERNION FOR THE     C

C                OPTIMAL ROTATION FROM THE NON-NORMALIZED ATTITUDE     C

C                PROFILE MATRIX                                        C

C                                                                      C

C                ANALYSIS FOR THIS ALGORITHM IS GIVEN IN               C

C                   M. D. SHUSTER, APPROXIMATE ALGORITHMS FOR          C

C                   FAST OPTIMAL ATTITUDE COMPUTATION, PAPER NO.       C

C                   78-1249, AIAA GUIDANCE AND CONTROL CONFER-         C

C                   ENCE, PALO ALTO, CALIF., AUGUST 1978               C

C                                                                      C

C-----ARGUMENT LIST--                                                  C

C                                                                      C

C     NAME          TYPE I/O     DESCRIPTION                           C

C     -----         ---- ---     -----------                           C

C                                                                      C

C     B(3,3)        R*8   I      NON-NORMALIZED ATTITUDE PROFILE       C

C                                MATRIX                                C

C                                                                      C

C     SIGTM2        R*8   I      SUMMATION OF INVERSE SQUARE OF THE    C

C                                STANDARD DEVIATIONS PER AXIS OF UNIT  C

C                                VECTOR MEASUREMENTS                   C

C                                                                      C

C     NEWT          I*4   I      NUMBER OF NEWTON-RAPHSON ITERATIONS   C

C                                                                      C

C     QUETCH        R*8   I      DESIRED ACCURACY IN DETERMINING       C

C                                OVERLAP EIGENVALUE (ALAM)             C

C                                                                      C

C     Q(4)          R*8   O      OUTPUT QUATERNION (UNNORMALIZED)      C

C                                                                      C

C     ALAM          R*8   O      OVERLAP EIGENVALUE (IF NEWT=0,        C

C                                THEN ALAM=1.D0)                       C

C                                                                      C

C     QUAL          R*8   O      QUAL = DABS(Q(4)) IS A MEASURE OF     C

C                                THE ACCURACY IN COMPUTING Q. QUAL     C

C                                IS LARGER THE MORE ACCURATE IS Q.     C

C                                                                      C

C-----EXTERNAL REFERENCES--NONE                                        C

C                                                                      C

C-----CALLED FROM--BQUEST                                              C

C                                                                      C



C-----DESIGNER/PROGRAMMER--M. D. SHUSTER  C.S.C. 1/17/1979             C

C                                                                      C

C-----MODIFICATIONS-- D. G. KUBITSCHEK  JHU-APL  06/25/90              C

C                     DECLARE ALL VARIABLES REAL*8. ALLOW FOR INPUT    C

C                     OF NON-NORMALIZED ATTITUDE PROFILE MATRIX.       C

C                                                                      C

C----------------------------------------------------------------------C

C

C

      IMPLICIT REAL*8 (A-H,O-Z)

      DIMENSION B(3,3),Q(4),S(3,3),Z(3),SZ(3),SSZ(3)

C

C

      DO 6 I=1,3

      DO 5 J=1,3

      B(I,J)=B(I,J)/SIGTM2

    5 CONTINUE

    6 CONTINUE

C

C

C

C     COMPUTE MATRIX S = B + TRANSPOSE

C

C

      DO 20 I=1,3

      DO 10 J=1,3

      S(I,J) = B(I,J) + B(J,I)

   10 CONTINUE

   20 CONTINUE

C

C

C

C

C     COMPUTE TRACE, TRACE(ADJOINT), AND DETERMINANT OF S

C

      TRS = S(1,1)+S(2,2)+S(3,3)

C

      TRADJS = S(1,1)*S(2,2)+S(2,2)*S(3,3)+S(3,3)*S(1,1)

     1        -S(1,2)*S(2,1)-S(2,3)*S(3,2)-S(3,1)*S(1,3)

C

      DETS = S(1,1)*(S(2,2)*S(3,3)-S(2,3)*S(3,2))

     1      +S(1,2)*(S(2,3)*S(3,1)-S(2,1)*S(3,3))

     2      +S(1,3)*(S(3,2)*S(2,1)-S(3,1)*S(2,2))

C

C

      SIGMA = TRS/2.D0

C

C

C     COMPUTE VECTOR Z

C

      Z(1) = B(2,3) - B(3,2)

      Z(2) = B(3,1) - B(1,3)

      Z(3) = B(1,2) - B(2,1)

C

C

C     COMPUTE VECTORS SZ AND SSZ

C

      SZ(1) = S(1,1)*Z(1) + S(1,2)*Z(2) + S(1,3)*Z(3)

      SZ(2) = S(2,1)*Z(1) + S(2,2)*Z(2) + S(2,3)*Z(3)

      SZ(3) = S(3,1)*Z(1) + S(3,2)*Z(2) + S(3,3)*Z(3)

C

      SSZ(1) = S(1,1)*SZ(1) + S(1,2)*SZ(2) + S(1,3)*SZ(3)

      SSZ(2) = S(2,1)*SZ(1) + S(2,2)*SZ(2) + S(2,3)*SZ(3)

      SSZ(3) = S(3,1)*SZ(1) + S(3,2)*SZ(2) + S(3,3)*SZ(3)

C

C

C     SET ALAM (CALLED X) TO UNPERTURBED VALUE (PERFECT OVERLAP)

C



      X = 1.D0

C

C

C     IF NEWT IS GREATER THAN ZERO, COMPUTE NEW VALUE OF ALAM

C     USING NEWTON-RAPHSON METHOD NEWT TIMES. OTHERWISE, PROCEED

C     DIRECTLY TO THE COMPUTATION OF THE QUATERNION USING THE

C     STARTING VALUE FOR ALAM.

C

      IF(NEWT.LE.0)GO TO 40

C

C     DETERMINE COEFFICIENTS OF CHARACTERISTC POLYNOMIAL FOR ALAM

C                X**4 + D1*X**2 + D2*X + D3

C

      B1 = SIGMA**2 - TRADJS

      B2 = SIGMA**2 + Z(1)*Z(1) + Z(2)*Z(2) + Z(3)*Z(3)

      B3 = DETS + Z(1)*SZ(1) + Z(2)*SZ(2) + Z(3)*SZ(3)

      B4 = Z(1)*SSZ(1) + Z(2)*SSZ(2) + Z(3)*SSZ(3)

C

      D1 = -B1-B2

      D2 = -B3

      D3 = B1*B2+SIGMA*B3-B4

C

C     PERFORM NEWTON-RAPHSON ITERATIONS

C

      DO 30 I=1,NEWT

      X2 = X*X

      DX = -(((X2+D1)*X +D2)*X+D3)/((4.D0*X2+2.D0*D1)*X+D2)

      X = X+DX

C

C     TEST IF ALAM COMPUTATION EXCEEDS DESIRED ACCURACY.

C     IF SO, EXIT FROM LOOP

C

      IF(DABS(DX).LT.QUETCH)GO TO 40

C

   30 CONTINUE

C

C

C     COMPUTE UNNORMALIZED AXIS VECTOR

C

   40 ALPHA = X*X - SIGMA**2 + TRADJS

      BETA = X - SIGMA

      GAMMA = ALPHA*(X+SIGMA)-DETS

C

C     COMPUTE UNNORMALIZED QUATERNION

C

      Q(1) = ALPHA*Z(1) + BETA*SZ(1) + SSZ(1)

      Q(2) = ALPHA*Z(2) + BETA*SZ(2) + SSZ(2)

      Q(3) = ALPHA*Z(3) + BETA*SZ(3) + SSZ(3)

      Q(4) = GAMMA

C

C

C     SET VALUE OF OVERLAP EIGENVALUE AND QUALITY PARAMETER

C

      ALAM = X

C

      QUAL = DABS(GAMMA)

C

      DO 60 I=1,3

      DO 50 J=1,3

      B(I,J) = B(I,J)*SIGTM2

   50 CONTINUE

   60 CONTINUE

C

      RETURN

      END



C = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = =

      SUBROUTINE QUACOV(F,FIBBL,VARTOT,IDET,P)

C

C-----VERSION DATE- JUNE 25, 1990 -------------------------------------C

C                                                                      C

C-----FUNCTION-- QUACOV COMPUTES THE COVARIANCE MATRIX OF THE QUA-     C

C                TERNION COMPUTED BY QUEST                             C

C                                                                      C

C-----ARGUMENT LIST--                                                  C

C                                                                      C

C     NAME          TYPE I/O     DESCRIPTION                           C

C     ----          ---- ---     -----------                           C

C                                                                      C

C     F(3,3)        R*8   I      OBSERVABILITY MATRIX                  C

C                                                                      C

C     FIBBL         R*8   I      CONDITION LIMIT ON F                  C

C                                (IF DABS(DET F) IS LESS THAN FIDDL,   C

C                                THEN F IS ILL-CONDITIONED             C

C                                                                      C

C     VARTOT        R*8   I      TOTAL VARIANCE OF THE OBSERVATION     C

C                                VECTOR MEASUREMENTS (IN RAD**2)       C

C                                                                      C

C     IDET          I*4   O      CONDITION CODE                        C

C                                (IDET=0,IF F IS WELL-CONDITIONED      C

C                                (IDET=1,IF F IS ILL-CONDITIONED       C

C                                                                      C

C     P(3,3)        R*8   O      COVARIANCE MATRIX (UNITLESS)          C

C                                (DEFAULT VALUE = 999. IN EVERY        C

C                                ELEMENT)                              C

C                                                                      C

C-----EXTERNAL REFERENCES--NONE                                        C

C                                                                      C

C-----CALLED FROM--BQUEST                                              C

C                                                                      C

C-----DESIGNER/PROGRAMMER--M. D. SHUSTER  C.S.C. 1/17/1979             C

C                                                                      C

C-----MODIFICATIONS-- D. G. KUBITSCHEK  JHU-APL  06/26/90              C

C                     DECLARE ALL VARIABLES REAL*8.                    C

C                                                                      C

C----------------------------------------------------------------------C

C

C

      IMPLICIT REAL*8 (A-H,O-Z)

      DIMENSION F(3,3),P(3,3)

C

C

C     SET INITIAL VALUE OF IDET

C

      IDET = 0

C

C

C     COMPUTE DETERMINANT OF F

C

C

      DETF = F(1,1)*(F(2,2)*F(3,3)-F(2,3)*F(3,2))

     1      +F(1,2)*(F(2,3)*F(3,1)-F(2,1)*F(3,3))

     2      +F(1,3)*(F(3,2)*F(2,1)-F(3,1)*F(2,2))

C

C

C     TEST CONDITION OF F. IF F IS ILL-CONDITIONED, SET CONDITION

C     CODE, COMPUTE DEFAULT COVARIANCE MATRIX, AND RETURN. OTHER-

C     WISE, CONTINUE.

C



      IF(DABS(DETF).GT.FIBBL)GO TO 50

C

      IDET = 1

C

      DO 40 I=1,3

      DO 30 J=1,3

      P(I,J) = 999.D0

   30 CONTINUE

   40 CONTINUE

C

      GO TO 999

C

C

   50 FAC = VARTOT/(4.D0*DETF)

C

C

C     COMPUTE COVARIANCE MATRIX

C

      P(1,1) = FAC*(F(2,2)*F(3,3)-F(2,3)*F(3,2))

      P(1,2) = FAC*(F(1,3)*F(3,2)-F(1,2)*F(3,3))

      P(1,3) = FAC*(F(1,2)*F(2,3)-F(1,3)*F(2,2))

      P(2,1) = FAC*(F(2,3)*F(3,1)-F(2,1)*F(3,3))

      P(2,2) = FAC*(F(1,1)*F(3,3)-F(1,3)*F(3,1))

      P(2,3) = FAC*(F(2,1)*F(1,3)-F(2,3)*F(1,1))

      P(3,1) = FAC*(F(3,2)*F(2,1)-F(3,1)*F(2,2))

      P(3,2) = FAC*(F(3,1)*F(1,2)-F(1,1)*F(3,2))

      P(3,3) = FAC*(F(1,1)*F(2,2)-F(1,2)*F(2,1))

C

C

  999 RETURN

      END

C = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = = =

      SUBROUTINE VEX(V1,V2,V3,DV1,DV2,DV3,DNORM)

C

C-----VERSION DATE- JUNE 25, 1990 -------------------------------------C

C                                                                      C

C-----FUNCTION-- VEX NORMALIZES A GIVEN VECTOR. THE NULL VECTOR        C

C                IS RETURNED AS THE NULL VECTOR.                       C

C                                                                      C

C-----ARGUMENT LIST--                                                  C

C                                                                      C

C     NAME          TYPE I/O      DESCRIPTION                          C

C     ----          ---- ---     -----------                           C

C                                                                      C

C     V1,V2,V3      R*8   I      COMPONENTS OF THE INPUT VECTOR        C

C                                                                      C

C     DV1,DV2,DV3   R*8   O      COMPONENTS OF THE OUTPUT VECTOR       C

C                                                                      C

C     DNORM         R*8   O      NORM OF THE INPUT VECTOR              C

C                                                                      C

C-----EXTERNAL REFERENCES--NONE                                        C

C                                                                      C

C-----CALLED FROM--INQUEST                                             C

C                                                                      C

C-----DESIGNER/PROGRAMMER--M. D. SHUSTER  C.S.C.  6/20/79              C

C                                                                      C

C-----MODIFICATIONS-- D. G. KUBITSCHEK  JHU-APL  06/25/90              C

C                     DECLARE ALL VARIABLES REAL*8                     C

C                                                                      C

C----------------------------------------------------------------------C

C

      REAL*8 DV1,DV2,DV3,DNORM



C

      DV1 = V1

      DV2 = V2

      DV3 = V3

C

      DNORM = DSQRT(DV1**2 + DV2**2 + DV3**2)

C

      IF(DNORM.EQ.0.D0)GO TO 999

C

      DV1 = DV1/DNORM

      DV2 = DV2/DNORM

      DV3 = DV3/DNORM

 999  RETURN

      END


